DYNARE COURSE
Amsterdam University
Optimal policy and deterministic shocks

Michel Juillard

October 24, 2008

Optimal policy

Introduction

Dynare currently implements two manners to compute optimal policy in DSGE models

- optimal simple rules
- optimal rule under commitment (Ramsey policy)

Optimal simple rule

Exemple (Clarida, Gali, Gertler)

$$\begin{aligned} y_t &= \delta y_{t-1} + (1-\delta)E_t y_{t+1} + \sigma(r_t - E_t inf_{t+1}) + e_{y_t} \\ inf_t &= \alpha inf_{-1} + (1-\alpha)E_t inf_{t+1} + \kappa y_t + e_{inf_t} \\ r_t &= \gamma_1 inf_t + \gamma_2 y_t \end{aligned}$$

Objectif

$$\arg\min_{\gamma_1,\gamma_2} \text{var}(y) + \text{var}(inf)$$

$$= \arg\min_{\gamma_1,\gamma_2} \lim_{\beta\to 1} E_0 \sum_{t=1}^{\infty}(1-\beta)\beta^t(y_t^2 + inf_t^2)$$

DYNARE example

```
var y inf r;
varexo e_y e_inf;

parameters delta sigma alpha kappa gamma1 gamma2;

delta =  0.44;
kappa =  0.18;
alpha =  0.48;
sigma = -0.06;
```

(continued)

```
model(linear);
y  = delta*y(-1)+(1-delta)*y(+1)+sigma *(r-inf(+1))+e_y;
inf = alpha*inf(-1)+(1-alpha)*inf(+1)+kappa*y+e_inf;
r = gamma1*inf+gamma2*y;
end;

shocks;
var e_y;
stderr 0.63;
var e_inf;
stderr 0.4;
end;
```

(continued)

```
optim_weights;
inf 1;
y 1;
end;

gamma1 = 1.1;
gamma2 = 0;

osr_params gamma1 gamma2;

osr;
```

Another example

$$
\begin{aligned}
y_t &= \delta y_{t-1} + (1-\delta)E_t y_{t+1} + \sigma(r_t - E_t inf_{t+1}) + e_{y_t} \\
inf_t &= \alpha inf_{-1} + (1-\alpha)E_t inf_{t+1} + \kappa y_t + e_{inf_t} \\
r_t &= \gamma_1 inf_t + \gamma_2 y_t \\
dr_t &= r_t - r_{t-1}
\end{aligned}
$$

Objectif

$$\min_{\gamma_1,\gamma_2} \text{var}(y) + \text{var}(inf) + 0.2\text{var}(dr)$$

Ramsey policy: General nonlinear case

$$\max_{\{x_t\}_{t=0}^{\infty}} E_1 \sum_{t=1}^{\infty} \beta^{t-1} U(y_t)$$

s.t.

$$E_t f(y_{t+1}, y_t, y_{t-1}, \varepsilon_t) = 0$$

$y_t \in R^n$: endogenous variables

$\varepsilon_t \in R^p$: stochastic shocks

and

$$f : R^{3n+p} \rightarrow R^m$$

There are $n - m$ free policy instruments.

Lagrangian

$$L = E_1 \sum_{t=1}^{\infty} \beta^{t-1} U(y_t) - \lambda_t' \left(f(y_{t+1}, y_t, y_{t-1}, \varepsilon_t) \right)$$

$$\begin{aligned} \frac{\partial L}{\partial y_1} &= E_1 \big[U_1(y_1) - \mu_1' f_2(y_2, y_1, y_0, \varepsilon_1) \\ &\quad - \beta \mu_2' f_3(y_3, y_2, y_1, \varepsilon_2) \\ \frac{\partial L}{\partial y_t} &= E_1 \big[U_1(y_t) - \mu_t' f_2(y_{t+1}, y_t, y_{t-1}, \varepsilon_t) \\ &\quad - \beta \mu_{t+1}' f_3(y_{t+2}, y_{t+1}, y_t, \varepsilon_{t+1}) \\ &\quad - \beta^{-1} \mu_{t-1}' f_1(y_t, y_{t-1}, y_{t-2}, \varepsilon_{t-1}) \big] \quad t \geq 2 \end{aligned}$$

with $\mu_t = \lambda_t / \beta^{t-1}$

First order conditions

The first order conditions of this optimization problem are

$$\begin{aligned} E_t \big[U_1(y_t) - \mu_t' f_2(y_{t+1}, y_t, y_{t-1}, \varepsilon_t) & \\ - \beta \mu_{t+1}' f_3(y_{t+2}, y_{t+1}, y_t, \varepsilon_{t+1}) & \\ - \beta^{-1} \mu_{t-1}' f_1(y_t, y_{t-1}, y_{t-2}, \varepsilon_{t-1}) \big] &= 0 \\ E_t \big[f(y_{t+1}, y_t, y_{t-1}, \varepsilon_t) \big] &= 0 \end{aligned}$$

with $\mu_0 = 0$ and where $U_1()$ is the Jacobian of function $U()$ with respect to y_t and $f_i()$ is the first order partial derivative of $f()$ with respect to the ith argument. It is convenient to redefine the multipliers as $\mu_t = \lambda_t / \beta^{t-1}$.

Nature of the solution

The above system of equations is nothing but a larger system of nonlinear rational expectation equations. As such, it can be approximated either to first order or to second order. The solution takes the form

$$\begin{bmatrix} y_t \\ \mu_t \end{bmatrix} = \hat{g}\left(y_{t-2}, y_{t-1}, \mu_{t-1}, \varepsilon_{t-1}, \varepsilon_t \right)$$

The optimal policy is then directly obtained as part of the set of $g()$ functions.

Cautionary remark

The First Order Conditions for optimality are only necessary conditions for a maximum. Benigno and Woodford (2006) and Levine, Pearlman and Pierse (2007) propose algorithms to check a sufficient condition.

The steady state problem

The steady state is solution of

$$\begin{aligned} U_1(\bar{y}) - \bar{\mu}'\left[f_2(\bar{y},\bar{y},\bar{y},0) - \beta f_3(\bar{y},\bar{y},\bar{y},0)\right. & \\ \left. -\beta^{-1} f_1(\bar{y},\bar{y},\bar{y},0)\right] &= 0 \\ f(\bar{y},\bar{y},\bar{y},0) &= 0 \end{aligned}$$

Computing the steady state

For a given value $\tilde{y}$, it is possible to use the first matrix equation above to obtain the value of $\tilde{\mu}$ that minimizes the sum of square residuals, e:

$$\begin{aligned} M &= f_2(\tilde{y},\tilde{y},\tilde{y},\tilde{x},0) - \beta f_3(\tilde{y},\tilde{y},\tilde{y},\tilde{x},0) \\ &\quad -\beta^{-1} f_1(\tilde{y},\tilde{y},\tilde{y},0) \\ \tilde{\mu}' &= U_1(\tilde{y}) M \left(M'M\right)^{-1} \\ e' &= U_1(\tilde{y}) - \tilde{\mu}' M \end{aligned}$$

Furthermore, $\tilde{y}$ must satisfy the m equations

$$f(\tilde{y},\tilde{y},\tilde{y},0) = 0$$

It is possible to build a sytem of equations whith only n unkowns $\tilde{y}$, but we must provide $n-m$ independent measures of the residuals e. Independent in the sense that the derivatives of these measures with respect to $\tilde{y}$ must be linearly independent.

A QR trick

At the steady state, the following must hold exactly

$$U_1(\tilde{y}) = \bar{\mu}' M$$

This can only be if

$$M^\star = \begin{bmatrix} M \\ U_1(\tilde{y}) \end{bmatrix}$$

is of rank m The reordered QR decomposition of $M^\star$ is such that

$$\underset{(m+1)\times n}{M^\star} \quad \underset{n\times n}{E} = \underset{(m+1)\times(m+1)}{Q} \quad \underset{(m+1)\times n}{R}$$

where E is a permutation matrix, Q an orthogonal matrix and R a triangular matrix with diagonal elements ordered in decreasing size.

A QR trick (continued)

- When $U_1(\tilde{y}) = \bar{\mu}'M$ doesn't hold exactly $M^\star$ is full rank $(m+1)$ and the $n-m$ last elements of R may be different from zero.
- When $U_1(\tilde{y}) = \bar{\mu}'M$ holds exactly $M^\star$ has rank m and the $n-m$ last elements of R are zero.
- The last $n-m$ elements of the last row of R provide the $n-m$ independent measures of the residuals e
- In practice, we build a nonlinear function with $\tilde{y}$ as input and that returns the $n-m$ last elements of the last row of R and $f(\tilde{y},\tilde{y},\tilde{y},0)$. At the solution, when $\tilde{y}=\bar{y}$, this function must return zeros.

First order approximation of the FOCs

$$
\begin{aligned}
E_t\Big[U_{11}\widehat{y}_t - \widehat{\mu}_t' f_2 - \beta\widehat{\mu}_{t+1}' f_3 - \beta^{-1}\widehat{\mu}_{t-1}' f_1 - \bar{\mu}'[\beta f_{31}\left(I\otimes\widehat{y}_{t+2}\right) \\
+\beta^{-1}f_{13}\left(I\otimes\widehat{y}_{t-2}\right) + (f_{12}+\beta f_{32})\left(I\otimes\widehat{y}_{t+1}\right) \\
+(f_{23}+\beta^{-1}f_{12})\left(I\otimes\widehat{y}_{t-1}\right) + (f_{33}+\beta f_{22}+\beta^{-1}f_{11})\left(I\otimes\widehat{y}_t\right) \\
+f_{24}\left(I\otimes\widehat{\varepsilon}_t\right) + \beta f_{34}\left(I\otimes\widehat{\varepsilon}_{t+1}\right) + \beta^{-1}f_{14}\left(I\otimes\widehat{\varepsilon}_{t-1}\right])\Big] &= 0 \\
E_t\left[f_1\widehat{y}_{t+1} + f_2\widehat{y}_t + f_3\widehat{y}_{t-1} + f_4\widehat{\varepsilon}_t\right] &= 0
\end{aligned}
$$

where I is an identity matrix of suitable order, $\widehat{y}_t = y_t - \bar{y}$, $\widehat{\mu}_t = \mu_t - \bar{\mu}$ and f_{ij} indicates a matrix of second order derivatives corresponding to the ith and the jth argument of the $f()$ function.

Representation of 2nd order derivatives

the Hessians of each $f^i()$ function are unrolled across the columns of the matrix:

$$
f_{ij} = \left[\begin{array}{cccccc} \frac{\partial f}{\partial x_1 \partial x_1'} & \frac{\partial f}{\partial x_1 \partial x_2'} & \cdots & \frac{\partial f}{\partial x_1 \partial x_n'} & \frac{\partial f}{\partial x_2 \partial x_1'} & \cdots & \frac{\partial f}{\partial x_n \partial x_n'} \end{array}\right]
$$

Here, x_i, $(i=1,\dots,4)$, represents one of the 4 vector arguments of the function $f()$.

The first pitfall

A naive approach ot linear-quadratic appoximation that would consider a linear approximation of the dynamics of the system and a second order approximation of the objective function, ignores the second order derivatives f_{ij} that enter in the first order approximation of the dynamics of the model under optimal policy.

The approximated solution function

$$
\begin{aligned}
y_t &= \bar{y} + g_1 \widehat{y}_{t-2} + g_2 \widehat{y}_{t-1} + g_3 \widehat{\mu}_{t-1} + g_4 \varepsilon_{t-1} + g_5 \varepsilon_t \\
\mu_t &= \bar{\mu} + h_1 \widehat{y}_{t-2} + h_2 \widehat{y}_{t-1} + h_3 \widehat{\mu}_{t-1} + h_4 \varepsilon_{t-1} + h_5 \varepsilon_t
\end{aligned}
$$

Drawback from this approach

In first period, the initial condition is $\mu_0 = 0$ and the state of the sytem is given by y_0. However the suggested dependency of first period decision on ε_0 and y_{-1} is spurious. It comes from the fact that the first order approximation of

$$
\mu'_{t-1} f_1(y_t, y_{t-1}, y_{t-2}, \varepsilon_{t-1})
$$

is given by

$$
\begin{aligned}
&\bar{\mu} f_1 + \widehat{\mu}_{t-1} f_1 + \bar{\mu}\left(f_{11}\left(I \otimes \widehat{y}_t\right) + f_{12}\left(I \otimes \widehat{y}_{t-1}\right) + f_{13}\left(I \otimes \widehat{y}_{t-2}\right)\right. \\
&\left.+ f_{14}\left(I \otimes \varepsilon_{t-1}\right)\right)
\end{aligned}
$$

that doesn't vanishes for $\mu_0 = 0$.
It may be better to solve first the continuation problem for $t > 2$, then the first period problem.

A second order approximation of the Lagrangian

Computing a first order approximation of the FOCs is equivalent to computing the solution of a second order approximation of the Lagrangian:

$$
\begin{aligned}
&E_1 \sum_{t=1}^{\infty} \beta^{t-1} \Big(\bar{U} + U_1 \widehat{y}_t + \frac{1}{2} U_{11} \left(\widehat{y}_t \otimes \widehat{y}_t\right) \\
&- \bar{\mu}' \Big(\bar{f} + f_1 \widehat{y}_{t+1} + f_2 \widehat{y}_t + f_3 \widehat{y}_{t-1} + f_4 \varepsilon_t \\
&+ \frac{1}{2} \left(f_{11} \left(\widehat{y}_{t+1} \otimes \widehat{y}_{t+1}\right) + f_{22} \left(\widehat{y}_t \otimes \widehat{y}_t\right) + f_{33} \left(\widehat{y}_{t-1} \otimes \widehat{y}_{t-1}\right) + f_{44} \left(\varepsilon_t \otimes \varepsilon_t\right) \right) \\
&+ f_{12} \left(\widehat{y}_{t+1} \otimes \widehat{y}_t\right) + f_{13} \left(\widehat{y}_{t+1} \otimes \widehat{y}_{t-1}\right) + f_{14} \left(\widehat{y}_{t+1} \otimes \varepsilon_t\right) + f_{23} \left(\widehat{y}_t \otimes \widehat{y}_{t-1}\right) \\
&+ f_{24} \left(\widehat{y}_t \otimes \varepsilon_t\right) + f_{34} \left(\widehat{y}_{t-1} \otimes \varepsilon_t\right) \Big) - \widehat{\mu}'_t \bar{f} \\
&- \widehat{\mu}'_t \left(f_1 \widehat{y}_{t+1} + f_2 \widehat{y}_t + f_3 \widehat{y}_{t-1} + f_5 \varepsilon_t \right) \Big)
\end{aligned}
$$

with $\bar{U} = U(\bar{y})$ and $\bar{f} = f(\bar{y}, \bar{y}, \bar{y}, 0)$

Simplifying

From steady state definition

$$
\bar{f} = 0
$$

From first order conditions

$$
E_t \left[f_1 \widehat{y}_{t+1} + f_2 \widehat{y}_t + f_3 \widehat{y}_{t-1} + f_4 \widehat{\varepsilon}_t \right] = 0
$$

We get

$$
\begin{aligned}
&E_1 \sum_{t=1}^{\infty} \beta^{t-1} \Big(\bar{U} + U_1 \widehat{y}_t + \frac{1}{2} U_{11} \left(\widehat{y}_t \otimes \widehat{y}_t\right) \\
&- \bar{\mu}' \Big(\frac{1}{2} \left(f_{11} \left(\widehat{y}_{t+1} \otimes \widehat{y}_{t+1}\right) + f_{22} \left(\widehat{y}_t \otimes \widehat{y}_t\right) + f_{33} \left(\widehat{y}_{t-1} \otimes \widehat{y}_{t-1}\right) \right. \\
&\left. + f_{44} \left(\varepsilon_t \otimes \varepsilon_t\right) \right) + f_{12} \left(\widehat{y}_{t+1} \otimes \widehat{y}_t\right) + f_{13} \left(\widehat{y}_{t+1} \otimes \widehat{y}_{t-1}\right) + f_{14} \left(\widehat{y}_{t+1} \otimes \varepsilon_t\right) \\
&f_{23} \left(\widehat{y}_t \otimes \widehat{y}_{t-1}\right) + f_{24} \left(\widehat{y}_t \otimes \varepsilon_t\right) + f_{34} \left(\widehat{y}_{t-1} \otimes \varepsilon_t\right) \Big) \Big)
\end{aligned}
$$

The second pitfall

- The FOCs set the first order Taylor expansion of $f()$ equal to zero, but not the second order terms
- Plugging the first order laws of motion for y_t into a second order approximation of the objective function ignores ignore the interaction between the Lagrange multipliers and the second order derivatives of the dynamic constraints.
- This equation must be approximated as such, it can't be computed together with a second order approximation of the FOCs: A second order approximation of the FOC bring in the third order derivatives of the objective function.

A recursive formula for computing welfare

Writing welfare in recusive form as

$$W_t = E_t\left(L_t + \beta W_{t+1}\right)$$

where L_t is one period expression for the Lagrangian. and replacing y_t, μ_t, y_{t+1}, μ_{t+1} and y_{t+2} by their expression as a function of the state variables, it is possible to write welfare as a function of the state:

$$W\left(y_{t-1}, \mu_{t-1}, \varepsilon_{t-1}, \varepsilon_t, \sigma^2\right)$$

where σ^2 is the stochastic scale of the system (as usual in a 2nd order approximation).
Note that in first period, μ_{t-1} should be set to zero.

Timeless perspective

- Time inconsistency problem stems from the fact that the initial value of the lagged Lagrange multipliers is set to zero.
- If, later on, the authorities re–optimize, they reset the Lagrange multipliers to zero.
- This mechanism reflects the fact that authorities make their decision after that private agents have formed their expectations (on the basis of the previous policy).
- When private agents expect the authorities to reoptimize the economy switch to a Nash equilibrium (discretionary solution)
- For optimal policy in a timeless perspective, Svensson and Woodford suggest that authorities relinquish their first period advantage and act as if the Lagrange mulitpliers had been initialized to zero in the far away past.
- What should be the initial value of the Lagrange multipliers for optimal policy in a timeless perspective?

Implementing timeless–perspective

First approach, (Juillard & Pelgrin, 2006):

1. Estimate model until date of implementation of optimal policy in a timeless–perspective with an empirical policy rule and compute smoothed value of unobservable variables.
2. Derive recursive solution for Ramsey policy rule (optimal policy under commitment).
3. Compute ex–post values for the Lagrange multipliers over the estimation sample, setting initial values of Lagrange multipliers at the beginning of estimation sample to zero.
4. Compute optimal policy in a timeless–perspective after the end of the estimation sample. Last values of Lagrange multipliers computed in Step 3 and smoothed variables provide initial values.

Second approach: Eliminating the Lagrange multipliers

A linear approximation of the solution to a Ramsey problem takes the form:

$$\begin{aligned} \begin{bmatrix} \widehat{y}_t \\ \widehat{\mu}_t \end{bmatrix} &= \begin{bmatrix} A_{11} & A_{12} \\ A_{21} & A_{22} \end{bmatrix} \begin{bmatrix} \widehat{y}_{t-1} \\ \widehat{\mu}_{t-1} \end{bmatrix} + \begin{bmatrix} B_1 \\ B_2 \end{bmatrix} \varepsilon_t \\ &= A_1 \widehat{y}_{t-1} + A_2 \widehat{\mu}_{t-1} + B\varepsilon_t \end{aligned}$$

where A_1 and A_2 are the conforming sub-matrices.
The problem is to eliminate $\widehat{\mu}_{t-1}$ from the expression for $\widehat{y}_t$ by substituting values of $\widehat{y}_{t-1}, \widehat{y}_{t-2}, \varepsilon_{t-1}$.

A QR algolrithm (I)

$$A_2 E = QR = \begin{bmatrix} Q_{11} & Q_{12} \\ Q_{21} & Q_{22} \end{bmatrix} \begin{bmatrix} R_1 & R_2 \\ 0 & 0 \end{bmatrix}$$

where E is a permutation matrix, R_1 is upper triangular and R_2 is empty if A_2 is full (column) rank.
Replacing A_2 by its decomposition, one gets

$$Q' \begin{bmatrix} \widehat{y}_t \\ \widehat{\mu}_t \end{bmatrix} = Q' A_1 \widehat{y}_{t-1} + RE' \widehat{\mu}_{t-1} + Q' B \varepsilon_t$$

The bottom part of the above system can be written

$$Q'_{12} \widehat{y}_t + Q'_{22} \widehat{\mu}_t = \begin{bmatrix} Q'_{12} & Q'_{22} \end{bmatrix} (A_1 \widehat{y}_{t-1} + B\varepsilon_t)$$

This system in turn contains necessarily more equations than elements in $\widehat{\mu}_t$.

A QR algolrithm (II)

A new application of the QR decomposition to Q'_{22} gives:

$$Q'_{22} \widetilde{E} = \widetilde{Q}\widetilde{R} = \begin{bmatrix} \widetilde{Q}_{11} & \widetilde{Q}_{12} \\ \widetilde{Q}_{21} & \widetilde{Q}_{22} \end{bmatrix} \begin{bmatrix} \widetilde{R}_1 & \widetilde{R}_2 \\ 0 & 0 \end{bmatrix}$$

where $\widetilde{E}$ is a permutation matrix. When R_2 isn't empty, the system is undetermined and it is possible to choose some of the multipliers μ_t, the ones corresponding to the columns of $\widetilde{R}_2$. We set them to zero, following a minimal state space type of arguments. Note however, that the QR decomposition isn't unique. Once the economy is managed according to optimal policy, all choices of decomposition would be equivalent, but the choice may matter for the initial period. This is an issue that deserves further studying.

A QR algolrithm (III)

Then, we have

$$\widehat{\mu}_t = \widetilde{E} \begin{bmatrix} \widetilde{R}_1^{-1} \begin{bmatrix} \widetilde{Q}'_{11} & \widetilde{Q}'_{21} \end{bmatrix} \left(\begin{bmatrix} Q'_{12} & Q'_{22} \end{bmatrix} (A_1 \widehat{y}_{t-1} + B\varepsilon_t) - Q'_{12} \widehat{y}_t \right) \\ 0 \end{bmatrix}$$

and

$$\widehat{\mu}_{t-1} = \widetilde{E} \begin{bmatrix} \widetilde{R}_1^{-1} \begin{bmatrix} \widetilde{Q}'_{11} & \widetilde{Q}'_{21} \end{bmatrix} \left(\begin{bmatrix} Q'_{12} & Q'_{22} \end{bmatrix} (A_1 \widehat{y}_{t-2} + B\varepsilon_{t-1}) - Q'_{12} \widehat{y}_{t-1} \right) \\ 0 \end{bmatrix}$$

A QR algolrithm (IV)

Replacing, $\widehat{\mu}_{t-1}$ in the original equation for $\widehat{y}_t$, one obtains finally

$$\widehat{y}_t = M_1 \widehat{y}_{t-1} + M_2 \widehat{y}_{t-2} + M_3 \varepsilon_t + M_4 \varepsilon_{t-1}$$

where

$$\begin{aligned}
M_1 &= A_{11} - A_{12}\widetilde{E} \begin{bmatrix} \widetilde{R}_1^{-1} \begin{bmatrix} \widetilde{Q}'_{11} & \widetilde{Q}'_{21} \end{bmatrix} Q'_{12} \\ 0 \end{bmatrix} \\
M_2 &= A_{12}\widetilde{E} \begin{bmatrix} \widetilde{R}_1^{-1} \begin{bmatrix} \widetilde{Q}'_{11} & \widetilde{Q}'_{21} \end{bmatrix} \begin{bmatrix} Q'_{12} & Q'_{22} \end{bmatrix} A_1 \\ 0 \end{bmatrix} \\
M_3 &= B_1 \\
M_4 &= A_{12}\widetilde{E} \begin{bmatrix} \widetilde{R}_1^{-1} \begin{bmatrix} \widetilde{Q}'_{11} & \widetilde{Q}'_{21} \end{bmatrix} \begin{bmatrix} Q'_{12} & Q'_{22} \end{bmatrix} B \\ 0 \end{bmatrix}
\end{aligned}$$

A linear–quadratic example (I)

```
var y inf r dr;
varexo e_y e_inf;

parameters delta sigma alpha kappa gamma1 gamma2;

delta =  0.44;
kappa =  0.18;
alpha =  0.48;
sigma = -0.06;
```

A linear–quadratic example(II)

```
model(linear);
y  = delta*y(-1)+(1-delta)*y(+1)+sigma *(r-inf(+1))+e_y
inf = alpha*inf(-1)+(1-alpha)*inf(+1)+kappa*y+e_inf;
dr = r - r(-1);
end;

shocks;
var e_y;
stderr 0.63;
var e_inf;
stderr 0.4;
end;
```

A linear–quadratic example(III)

```
planner_objective y^2 + inf^2 + 0.2*dr^2;

ramsey_policy(planner_discount=1);
```

Deterministic shocks

Structural change

- Most existing DSGE models have constant parameters and zero-mean shocks.
- Can't handle long run tendencies such as demographic change or policy shift.
- Expected change triggers anticipatory behaviour.
- Unexpected change only in a radical sense: its occurrence isn't even thought possible. Otherwise it must be modeled as a stochastic shock.
- Usually discussed only in deterministic models.

Formal treatment of structural change

- Representing structural change as deterministic exogenous variables
- Unexpected change simply shift the simulation but doesn't enter expectations
- Expected change is part of information set when decisions are taken

Proposed solution: add future values of deterministic exogenous variables to the list of state variables.

A general model

A DSGE model can be represented by a set of stochastic equations:

$$E_t \left\{ f(y_{t+1}, y_t, y_{t-1}, x_t, u_t) \right\} = 0$$

where y_t is the vector of endogenous variables in the model, x_t is a vector of exogenous deterministic variables. By assumption, these variables settle to a constant value after some horizon, $x_\tau = \bar{x}$ for $\tau > T_N$. u_t is a vector of stochastic shocks defined as $u_t = \sigma \epsilon_t$, whith

$$E\left\{\epsilon_t\right\} = 0 \qquad E\left\{\epsilon_t \epsilon_t^T\right\} = \Sigma_\epsilon.$$

σ is a stochastic scale factor.

Perturbation approach

Unknown decision functions:

$$y_t = g(y_{t-1}, x_t, \dots, x_{t+N}, u_t, \sigma)$$

Recover the Taylor expansion coefficients of the unknown decision functions $g(\dots)$ from the Taylor expansion of the structural model $E\left\{f(\dots)\right\} = 0$.
WARNING: the perturbation approach involves a local approximation that may not be satisfactory for large changes in exogenous variables

Deterministic steady state

$$\begin{aligned} f(\bar{y}, \bar{y}, \bar{y}, \bar{x}, 0) &= 0 \\ \bar{y} &= g(\bar{y}, \bar{x}, \dots, \bar{x}, 0, 0) \end{aligned}$$

Note that the deterministic steady state is computed at the value at which the exogenous variables finally settle.

The model as function of state variables

$$\begin{aligned} y_t &= g(y_{t-1}, x_t, \dots, x_{t+N}, u_t, \sigma) \\ y_{t+1} &= g(y_t, x_{t+1}, \dots, x_{t+N}, \bar{x}, u_{t+1}, \sigma) \\ &= g(g(y_{t-1}, x_t, \dots, x_{t+N}, u_t, \sigma), x_{t+1}, \dots, x_{t+N}, \bar{x}, u_{t+1}, \sigma) \end{aligned}$$

$$F(y_{t-1}, x_t, \dots, x_{t+N}, u_t, u_{t+1}, \sigma) = \\ f(g(g(y_{t-1}, x_t, \dots, x_{t+N}, u_t, \sigma), x_{t+1}, \dots, x_{t+N}, \bar{x}, u_{t+1}, \sigma), \\ g(y_{t-1}, x_t, \dots, x_{t+N}, u_t, \sigma), y_{t-1}, x_t, u_t, \sigma)$$

$$E_t\left\{F(y_{t-1}, x_t, \dots, x_{t+N}, u_t, u_{t+1}, \sigma)\right\} = 0$$

First order approximation

$$F^{(1)}(y_{t-1}, x_t, \dots, x_{t+N}, u_t, u_{t+1}, \sigma) = \\ F(\bar{y}, \bar{x}, \dots, \bar{x}, 0, 0) + F_y\hat{y} + F_{x_1}\hat{x}_1 + \dots + F_{x+N}\hat{x}_N + F_u u + F_{u'} u' + F_\sigma \sigma$$

with $\hat{y} = y_{t-1} - \bar{y}, \hat{x}_1 = x_t - \bar{x}, \dots, \hat{x}_N = x_{t+N} - \bar{x}, u = u_t, u' = u_{t+1}$.

$$\begin{aligned} & E_t\left\{F^{(1)}(y_{t-1}, \hat{x}_t, \dots, \hat{x}_{t+N}, u_t, u_{t+1}, \sigma)\right\} \\ &= F(\bar{y}, \bar{x}, \dots, \bar{x}, 0, 0) + F_y\hat{y} + F_{x_1}\hat{x}_1 + \dots + F_{x+N}\hat{x}_N + F_u u + F_\sigma \sigma \\ &= 0 \\ F_y &= 0 \\ F_u &= 0 \\ F_\sigma &= 0 \\ F_{x_1} &= 0 \\ \dots & \\ F_{x_N} &= 0 \end{aligned}$$

Recovering g_y, g_u, g_σ

- g_y is recovered from $F_y = 0$
- g_u is recovered from $F_u = 0$
- $g_\sigma (= 0)$ is recovered from $F_\sigma = 0$
- g_{x_1} is recovered form $F_{x_1} = 0 \ldots$
- g_{x_N} is recovered form $F_{x_N} = 0$

g_x

$$
\begin{aligned}
F_{x_1} &= f_+ g_y g_{x_1} + f_0 g_{x_1} + f_x \\
&= 0 \\
g_{x_1} &= -\left(f_+ g_y + f_0\right)^{-1} f_x \\
F_{x_i} &= f_+ \left(g_y g_{x_i} + g_{x_{i-1}}\right) + f_0 g_{x_i} & i = 2, \ldots, N \\
&= 0 \\
g_{x_i} &= -\left(f_+ g_y + f_0\right)^{-1} f_+ g_{x_{i-1}} & i = 2, \ldots, N
\end{aligned}
$$

Second order approximation

$$
\begin{aligned}
&F^{(2)}(y_{t-1}, x_t, \ldots, x_{t+N}, u_t, u_{t+1}, \sigma) = \\
&\quad F^{(1)}(y_{t-1}, x_t, \ldots, x_{t+N}, u_t, u_{t+1}, \sigma) + 0.5(F_{yy}(\hat{y} \otimes \hat{y}) + F_{uu}(u \otimes u) + F_{u'u'}(u' \otimes u') \\
&\quad + F_{\sigma\sigma}\sigma^2) + F_{yu}(\hat{y} \otimes u) + F_{yu'}(\hat{y} \otimes u') + F_{yx_1}(\hat{y} \otimes \hat{x}_1) + \ldots + F_{yx_N}(\hat{y} \otimes \hat{x}_N) \\
&\quad + F_{y\sigma}(\hat{y} \otimes \sigma) + F_{uu'}(u \otimes u') + F_{ux_1}(u \otimes \hat{x}_1) + \ldots + F_{ux_N}(u \otimes \hat{x}_N) + F_{u\sigma}(u \otimes \sigma) \\
&\quad + F_{u'x_1}(u' \otimes \hat{x}_1) + \ldots + F_{u'x_N}(u' \otimes \hat{x}_N) + F_{u'\sigma}(u' \otimes \sigma) + F_{x_1x_2}(\hat{x}_1 \otimes \hat{x}_2) \\
&\quad + \ldots + F_{x_{N-1}x_N}(\hat{x}_{N-1} \otimes \hat{x}_N) + F_{x_1\sigma}(\hat{x}_1\sigma) + \ldots + F_{x_N\sigma}(\hat{x}_N\sigma) \\
&E_t\left\{F^{(2)}(y_{t-1}, x_t, \ldots, x_{t+N}, u_t, u_{t+1}, \sigma)\right\} = \\
&\quad F^{(1)}(y_{t-1}, x_t, \ldots, x_{t+N}, u_t, u_{t+1}, \sigma) + 0.5(F_{yy}(\hat{y} \otimes \hat{y}) + F_{uu}(u \otimes u) + F_{u'u'}\sigma^2\vec{\Sigma}_\epsilon \\
&\quad + F_{\sigma\sigma}\sigma^2) + F_{yu}(\hat{y} \otimes u) + F_{yx_1}(\hat{y} \otimes \hat{x}_1) + \ldots + F_{yx_N}(\hat{y} \otimes \hat{x}_N) + F_{y\sigma}(\hat{y} \otimes \sigma) \\
&\quad + F_{ux_1}(u \otimes \hat{x}_1) + \ldots + F_{ux_N}(u \otimes \hat{x}_N) + F_{u\sigma}(u \otimes \sigma) + F_{x_1x_2}(\hat{x}_1 \otimes \hat{x}_2) \\
&\quad + \ldots + F_{x_{N-1}x_N}(\hat{x}_{N-1} \otimes \hat{x}_N) + F_{x_1\sigma}(\hat{x}_1\sigma) + \ldots + F_{x_N\sigma}(\hat{x}_N\sigma) \\
&= \quad 0
\end{aligned}
$$

Recovering g_{yx_i}, g_{ux_i}

$$
\begin{aligned}
F_{yx_1} &= f_+ g_y g_{yx_1} + f_0 g_{yx_1} + R_1 \\
&= 0 \\
g_{yx_1} &= -\left(f_+ g_y + f_0\right)^{-1} R_1 \\
F_{yx_i} &= f_+ \left(g_{yx_{i-1}}(g_y \otimes I) + g_y g_{yx_i}\right) + f_0 g_{yx_i} + R_i \\
&= 0 \\
g_{yx_i} &= -\left(f_+ g_y + f_0\right)^{-1} \left(f_+ g_{yx_{i-1}}(g_y \otimes I) + R_i\right) \\
& \quad i = 2, \ldots, N
\end{aligned}
$$

where the terms $R_1, \ldots, R_N$, don't contain second order derivatives of $g()$ with respect to x_1,.

g_{ux_i} is recovered in a similar manner from $F_{ux_i} = 0$.

Recovering $g_{x_i x_j}$

$$
\begin{aligned}
F_{x_1 x_1} &= f_+ g_y g_{x_1 x_1} + f_0 g_{x_1 x_1} + R_{11} \\
&= 0 \\
g_{x_1 x_1} &= -\left(f_+ g_y + f_0\right)^{-1} R_{11} \\
F_{x_1 x_i} &= f_+ \left(g_{y x_{i-1}}(g_{x_1} \otimes I) + g_y g_{x_1 x_i}\right) + f_0 g_{x_1 x_i} + R_{1i} \\
&= 0 \\
g_{x_1 x_i} &= -\left(f_+ g_y + f_0\right)^{-1} \left(f_+ g_{y x_{i-1}}(g_{x_1} \otimes I) + R_{1i}\right) \\
F_{x_i x_j} &= f_+ \left(g_{y x_{j-1}}(g_{x_i} \otimes I) + g_{x_{i-1} y}(I \otimes g_{x_j}) + g_y g_{x_i x_j} + g_{x_{i-1} x_{j-1}}\right) + f_0 g_{x_i x_j} + R_{ij} \\
&= 0 \\
g_{x_i x_j} &= -\left(f_+ g_y + f_0\right)^{-1} \left(f_+ \left(g_{y x_{j-1}}(g_{x_i} \otimes I) + g_{x_{i-1} y}(I \otimes g_{x_j}) + g_{x_{i-1} x_{j-1}}\right) + R_{ij}\right) \\
& \quad i = 2, \dots, N \qquad j = 2, \dots, N
\end{aligned}
$$

where the terms $R_{11}, \dots, R_{NN}$, don't contain second order derivatives of $g()$ with respect to $x_1, \dots, x_N$.

Approximated decision functions

$$
\begin{aligned}
y_t \approx\; & \bar{y} + 0.5 g_{\sigma\sigma} + g_y \hat{y} + g_u \hat{u} + g_{x_1} \hat{x}_1 + \dots + g_{x_N} \hat{x}_N + 0.5(g_{yy}(\hat{y} \otimes \hat{y}) \\
& + g_{uu}(u \otimes u) + g_{x_1 x_1}(\hat{x}_1 \otimes \hat{x}_1) + \dots + g_{x_N x_N}(\hat{x}_N \otimes \hat{x}_N)) + g_{yu}(\hat{y} \otimes u) \\
& + g_{y x_1}(\hat{y} \otimes x_1) + \dots + g_{y x_N}(\hat{y} \otimes x_N) + g_{u x_1}(u \otimes x_1) + \dots + g_{u x_N}(u \otimes x_N) \\
& + g_{x_1 x_2}(\hat{x}_1 \otimes x_2) + \dots + g_{x_{N-1} x_N}(\hat{x}_{N-1} \otimes x_N)
\end{aligned}
$$

Hairault, Langot and Portier (2001) model

Welfare:

$$W_t = \ln c_t + \eta \ln(1 - h_t) + \beta E_t \left\{W_{t+1}\right\}$$

with W, welfare, c, consumption, and h, labor effort. $\eta = 2$

Optimality condition for consumption

$$\frac{1}{c_t} = E_t \left\{\beta \frac{1}{c_{t+1}} (z_{t+1} + 1 - \delta)\right\}$$

where z is the rate of return on capital, net of taxes. $\beta = 0.988$, $\delta = 0.025$

Model (continued)

Optimality condition for labor effort

$$\frac{\eta}{1 - h_t} = \frac{\omega_t}{c_t}$$

where ω is the wage rate, net of taxes.

Net wage rate

$$(1 - \alpha) \left(\frac{k_{t-1}}{h_t}\right)^{\alpha} = (1 + \mu)(1 + \tau_t)\omega_t$$

where k_{t-1} is the stock of capital at the end of the previous period and τ is the tax rate. $\alpha = 0.36$, $\mu = 0.1$

Model (continued)

Net rate of return on capital

$$\alpha \left(\frac{k_{t-1}}{h_t}\right)^{\alpha-1} = (1+\mu)(1+\tau_t)z_t$$

Accumulation

$$i_t = k_t - (1-\delta)k_{t-1}$$

with i, investment. $\delta = 0.025$

Model (continued)

Goods market equilibium

$$c_t + i_t = A_t k_{t-1}^{\alpha} h_t^{1-\alpha}$$

Total factor productivity

$$\ln A_t = (1-\rho)\ln \bar{A} + \rho \ln A_{t-1} + e_t$$

where $\rho = 0.95$.

An announced change in the tax rate

1. The economy is at the deterministic steady state corresponding to a tax rate $\tau = -0.15$ (arbitrary initial state)
2. In period 1, it is announced that the tax rate will be moved to the optimal value $\tau = -\frac{\mu}{1+\mu} \approx -0.0909$ in period 10.

hlp1.mod (I)

```
var Welf w c h i k z A;
varexo_det tau;
varexo e;

parameters beta delta alpha mu eta rho Abar;
delta = 0.025;
eta = 2;
mu = 0.1;
alpha = 0.36;
rho = 0.95;
beta = 0.988;
Abar = 1;
```

hlp1.mod (II)

```
model;
Welf = log(c)+eta*log(1-h)+beta*Welf(+1);
c+i = A*k(-1)^alpha*h^(1-alpha);
i = k - (1-delta)*k(-1);
log(A) = (1-rho)*log(Abar)+rho*log(A(-1))+e;
1/c = beta*(1/c(+1))*(z(+1)+1-delta);
eta/(1-h) = w/c;
alpha*(k(-1)/h)^(alpha-1) = (1+mu)*(1+tau)*z;
(1-alpha)*(k(-1)/h)^alpha = (1+mu)*(1+tau)*w;
end;
```

hlp1.mod (III)

```
initval;
Welf = -100;
w = 0.5;
c = 0.6;
h = 0.3;
i = 0.4;
k = 3;
z = 0.1;
A = 1;
tau = -0.15;
end;

steady;
```

hlp1.mod (IV)

```
endval;
Welf = -100;
w = 0.5;
c = 0.6;
h = 0.3;
i = 0.4;
k = 3;
z = 0.1;
A = 1;
tau = -mu/(1+mu);
end;

steady;
```

hlp1.mod (V)

```
shocks;
var e; stderr 0.01;
var tau;
periods 1:9;
values -0.15;
end;

stoch_simul(irf=0);

forecast(periods=40);
```